HORTSCIENCE 27(3):275. 1992.

Analysis of Pressure-volume Data Using Segmented, Nonlinear Regression Algorithms

Thomas G. Ranney
Department of Horticultural Science, Mountain Horticultural Crops Research and Extension Center, North Carolina State University, 2016 Fanning Bridge Road, Fletcher, NC 28732

David Davidson
Department of Floriculture and Ornamental Horticulture, Cornell University, Ithaca, NY 14853

Additional index words. SAS, Statistical Analysis System, osmotic potential, water potential isotherms, water relations

Pressure-volume methodology has been used extensively to study water relations of plant tissues. Measurements of tissue water potential (Ψ) are typically collected over a range of relative water contents *(R)* from saturation to well below the point of turgor loss. The methodology for collecting these data and the theoretical basis for interpreting the resulting curves have been discussed by Cheung et al. (1975).

Recently, nonlinear regression algorithms have been presented for fitting the entire pressure-volume curve as opposed to just the region below the point of turgor loss (Schulte and Hinckley, 1985). Herein we present one approach for analyzing pressure-volume data using a segmented, nonlinear regression program specifically written for the Statistical Analysis System (SAS, SAS Institute, Inc., Caxy, N.C.). The NLIN procedure in SAS is well suited for segmented, nonlinear regression analysis.

Use of a nonlinear, regression algorithm for fitting of curves requires selection of an appropriate mathematical model. Given that $\Psi = \psi_\pi + \psi_p$, Ψ can be described by the sum of its components, osmotic potential (ψ_π) and ψ_p, and each as a function of *R*. The osmotic component can be defined, based on theoretical considerations, and has been given by Schulte and Hinckley (1985): $\psi_\pi = \psi_{\pi,sat}/(1 - ((1 - R/X))$, where $\psi_{\pi,sat}$ is the osmotic potential at full saturation and X is the symplastic fraction of water at saturation.

The ψ_π component is more difficult to define because ψ_p is influenced by tissue elasticity and *R* and their relationships are somewhat unclear (Cheung et al., 1976). Various exponential functions, however, have been found to provide good empirical fits to pressure-volume data (Cheung et al., 1976).

Schulte and Hinckley (1985) have presented the modified exponential function: $\psi_p = \exp^{a(R - Ro)} - 1$, where "a" is a constant and *R*, is the *R* at the point of turgor loss. Modulus of elasticity, defined in terms of *R*, has been derived by Schulte and Hinckley (1985) as $E = (R - R_a)\, d\psi_p/dR$, where R_a is the apoplastic water fraction at saturation and is equal to 1 – X. When using the modified exponential equation for ψ_p, differentiation of *E* will give: *E* = a + a&. The maximum *E* (determined at full turgor) can be easily calculated using the differentiated equation for ***E*** and by substituting $-\psi_{\pi,sat}$ for ψ_p.

A SAS program, with the general model $\Psi = \psi_\pi + \psi_p$, using the aforementioned equations for ψ_π and ψ_π, is given as one approach for analyzing pressure-volume data (Fig. 1). However, different equations for Ψ_p may provide better fits depending on the species and tissue type. Precautions and problems associated with nonlinear regression analysis are discussed by Bates and Watts (1988) and SAS (1988).

Users should be aware that the shape of the *E* vs. ψ_p relationship depends on the model selected for ψ_p (Schulte and Hinckley, 1985). Thus, for certain studies where *E* and its relationship to other variables are of primary interest, care should be taken in selecting models and interpreting their output.

Copies of this and other programs (e.g., a power function) for analyzing pressure-volume data are available on diskette by sending a formatted DOS diskette to T.G.R. at the above address.

```
TITLE 'MODIFIED EXPONENTIAL MODEL';
DATA MEXPO;
    INFILE 'A:PV.DTA';
    INPUT WP FWT SATWT DRYWT;
    /* WP = water potential (a negative value), FWT = tissue weight at that WP,
    SATWT = saturated tissue weight, and DRYWT = tissue dry weight*/;
    R=(FWT-DRYWT)/(SATWT-DRYWT); WSD=1-R;
    /*R = relative water content and WSD = water saturation deficit*/;
PROC NLIN BEST=1 METHOD=MARQUARDT;
    BOUNDS SIPISAT<0, A>0, 1>X>0, 1>RTLP>0;
    PARMS  SIPISAT = -2.8 TO -1.0  BY .2  X = .6 TO 1.0 BY .1  A = 2 TO 12 BY 2
    RTLP = .80 TO .95 BY .025;
    /*Initial grid search.  Ranges should straddle expected values. SIPISAT = osmotic potential at full
    turgor (ψπ,sat), X = symplastic water fraction at saturation, A = constant, and RTLP = relative
    water content at the point of turgor loss (R0)*/;
    TURGOR=EXP(A*(R-RTLP))-1;
    SIPI=SIPISAT/(1-((1-R)/X));
    IF R>RTLP THEN DO;
        MODEL WP=SIPI+TURGOR;
        DER.SIPISAT=1/(1-((1-R)/X));
        DER.A=(R-RTLP)*EXP(A*(R-RTLP));
        DER.RTLP=-A*EXP(A*(R-RTLP));
        DER.X =-(SIPISAT*(1-R))/((1-((1-R)/X))**2*X**2);
    END;
    ELSE DO;
        MODEL WP=SIPI;
        DER.SIPISAT=1/(1-((1-R)/X));
        DER.X= -(SIPISAT*(1-R))/((1-((1-R)/X))**2*X**2);
    END;
    OUTPUT OUT = BOZO P=YHAT R=YRESID PARMS = SIPISAT X A RTLP;
PROC PLOT uniform vpercent=200; PLOT WP*WSD='*' YHAT*WSD='P'/overlay;
    PLOT YRESID*WSD/VREF=0;
DATA GONZO; SET BOZO;
    EMAX=A+(A * -SIPISAT); WPTLP=SIPISAT/(1-((1-RTLP)/X));
    SIPI=SIPISAT/(1-((1-R)/X)); TURGOR=WP-SIPI;
    /* WPTLP = water potential at the point of turgor loss and
    and EMAX = maximum bulk modulus of elasticity*/
PROC PRINT; VAR WP WSD SIPI TURGOR SIPISAT X A RTLP WPTLP EMAX;
RUN;
```

Fig. 1. A PC SAS program for analyzing pressure-volume data using the NLIN procedure with a modified exponential model for fitting the positive turgor portion of the curve.

Literature Cited

Bates, D.M. and D.G. Watts. 1988. Nonlinear regression analysis and its applications. Wiley, New York.

Cheung, Y.N.S M.T. Tyree, and J. Dainty. 1975. Water relations parameters on single leaves obtained in a pressure bomb and some ecological interpretations. Can. J. Bot. 53:1342-1346.

Cheung, Y.N.S., M.T. Tyree, and J. Dainty. 1976. Some possible sources of error in determining bulk elastic moduli and other parameters from pressure-volume curves of shoots and leaves. Can. J. Bot. 54:758-165.

SAS Institute, Inc. 1988. SAS/STAT user's guide: Release 6.03 edition. SAS Institute, Gary, N.C.

Schulte, P.J. and T.M. Hinckley. 1985. A comparison of pressure-volume curve data analysis techniques. J. Expt. Bot. 36:1590-1602.

Received for publication 28 Jan. 1991. Accepted for publication 17 Oct. 1991. The work reported in this publication was supported, in part, by the North Carolina Agricultural Research Service. The cost of publishing this paper was defrayed in part by the payment of page charges. Under postal regulations, this paper therefore must be hereby marked *advertisement* solely to indicate this fact.